
Change 1
	Function name
	fl_TC_11_2_4_Body ()

	Reason for change
	The current implementation of emergency attach in test case 11.2.4 calls the function f_EUTRA_NAS_AttachComplete with 2nd parameter set to tsc_EpsDefaultBearerId (=5), this is the bearer id that will be expected in ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT from the UE. But if a UE is configured for multi-PDN operation with IMS as the 2nd PDN, the id 12 (tsc_EpsDefaultBearerId2ndPDN) would actually have been used for the default bearer (this is determined inside f_EUTRA_IdleUpdated_Step14_15() using function f_EUTRA_InitialRegistrationEpsBearerId()).


	Summary of change
	Derive the bearer id using the function f_EUTRA_InitialRegistrationEpsBearerId(), to ensure it matches the value that would have been sent in the default bearer activation request.

	TTCN module
	LTE\11\IMS_EmergencyCall.ttcn

	MCC160 Comment
	Accepted in principle (see below)


Before change

function fl_TC_11_2_4_Body()runs on EUTRA_PTC

  {

    const UInt8_Type tsc_MaxEmerNumList := 7;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var float v_T3412Value := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 360.0); // 6min

    var template (value) GPRS_Timer v_T3412 := cs_GprsTimer_v(tsc_GprsTimerUnit_6min, '00001'B); // 6min

    var integer i;

    var CharStringList_Type v_EmergencyNumList;

     timer t_WatchDog := 5.0;
…

// Steps 5a1 to 15 of the generic test procedure for IMS Emergency call establishment in EUTRA: Limited Service (TS 36.508 4.5A.5.3-1).  @sic R5-142979 sic@

    //@siclog "Step 6-14" siclog@

    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps5a1_15(eutra_CellA, // @sic R5-134733r1 sic@ @sic R5s130989 sic@

                                                           v_NasInd,

                                                           -,

                                                           false,

                                                           v_T3412);  // @sic R5s130758 sic@

    // EXCEPTION: (IP address allocation), steps 1 - 5 of annex C.22 should be performed in parallel

    //@siclog "Step 15-16" siclog@

    // Void //@sic R5-134733r1 sic@

    //@siclog "Step 17" siclog@

    f_EUTRA_NAS_AttachComplete (eutra_CellA, v_ tsc_EpsDefaultBearerId, dontWaitForIMS);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17");

    // The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST with

    // particular settings (use Reference dedicated EPS bearer context #1 - see table 6.6.2-1 in TS 36.508) (See Note 1).

    //@siclog "Step 18-19" siclog@

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 20" siclog@

After change

     function fl_TC_11_2_4_Body()runs on EUTRA_PTC

  {

    const UInt8_Type tsc_MaxEmerNumList := 7;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var float v_T3412Value := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 360.0); // 6min

    var template (value) GPRS_Timer v_T3412 := cs_GprsTimer_v(tsc_GprsTimerUnit_6min, '00001'B); // 6min

    var integer i;

    var CharStringList_Type v_EmergencyNumList;

     var EPS_BearerIdentity v_EpsBearerId;
    timer t_WatchDog := 5.0;

…

// Steps 5a1 to 15 of the generic test procedure for IMS Emergency call establishment in EUTRA: Limited Service (TS 36.508 4.5A.5.3-1).  @sic R5-142979 sic@

    //@siclog "Step 6-14" siclog@

    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps5a1_15(eutra_CellA, // @sic R5-134733r1 sic@ @sic R5s130989 sic@

                                                           v_NasInd,

                                                           -,

                                                           false,

                                                           v_T3412);  // @sic R5s130758 sic@

    // EXCEPTION: (IP address allocation), steps 1 - 5 of annex C.22 should be performed in parallel

    //@siclog "Step 15-16" siclog@

    // Void //@sic R5-134733r1 sic@

    v_EpsBearerId := f_EUTRA_InitialRegistrationEpsBearerId();
    //@siclog "Step 17" siclog@

    f_EUTRA_NAS_AttachComplete (eutra_CellA, v_EpsBearerId, dontWaitForIMS);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17");

    // The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST with

    // particular settings (use Reference dedicated EPS bearer context #1 - see table 6.6.2-1 in TS 36.508) (See Note 1).

    //@siclog "Step 18-19" siclog@

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 20" siclog@

MCC160 Implementation
	  function f_EUTRA_InitialRegistrationPdnIndex(EUTRA_InitialRegistrationPDN_Type p_InitialRegistrationPDN) return PDN_Index_Type

  {

    var PDN_Index_Type v_PdnIndex;

    if (p_InitialRegistrationPDN == emergencyPDN) { /* @sic R5s150510 sic@ */

      v_PdnIndex := PDN_1;

    } 

    else {
      if (not tsc_EUTRA_Registration_MultiplePDN or

        ((p_InitialRegistrationPDN == firstPDN) and tsc_EUTRA_Registration_PDN1isIMS) or

        ((p_InitialRegistrationPDN == secondPDN) and tsc_EUTRA_Registration_PDN2isIMS)) {

      v_PdnIndex := PDN_1;

      } else {

        v_PdnIndex := PDN_Internet;

      }
    }
    return v_PdnIndex;

  }

	  function f_EUTRA_IdleUpdated_Step14_15(EUTRA_CellId_Type p_CellId,

                                         EUTRA_ATTACH_Type p_ForcedAttach,

                                         template (omit) AdditionalUpdateType p_AdditionalUpdateType,

                                         ProcedureTransactionIdentifier p_EPS_TI,

                                         boolean p_IPv4AllocationViaNasFlag,

                                         template (omit) ProtocolConfigOptions p_PcoToUE,

                                         AccessPointName p_APN,

                                         PDN_TypeToBeUsed_Type p_PDN_TypeToBeUsed := pdnTypeAsSupportedByUE,

                                         template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v_deact,

                                         template (omit) GPRS_Timer p_T3402 := omit,

                                         template (omit) GPRS_Timer p_T3423 := omit,

                                         template (omit) PLMN_List p_EquivalentPlmnList := omit,

                                         integer p_NoOfEmergencyNumbers := 0, // @sic R5-115673 sic@

                                         template (omit) EPS_NetworkFeatureSupport p_NtwkFeatSupport := cs_EPS_NwkFtSup_IMSVoice,

                                         template (omit) EMM_Cause p_EMM_Cause := omit,

                                         EUTRA_InitialRegistrationPDN_Type p_InitialRegistrationPDN := firstPDN) runs on EUTRA_PTC

  { /* @sic R5-112600: IP address allocation sic@ */

    /* @sic R5-110796: split into f_EUTRA_IdleUpdated_Step5_13 and f_EUTRA_IdleUpdated_Step14_15 sic@ */

    /* @sic R5-120234, R5-120541, R5-120542 sic@ */

    /* @sic R5-144370, R5-144797: p_APN shall not be omit sic@ */

    /* @sic R5-145971: new parameter p_EMM_Cause sic@ */

    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(p_ForcedAttach); // @sic R5-103673, R5s100795 sic@

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(p_CellId);

    var template (value) MobileIdentity v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode(p_CellId);

    var template (omit) LocAreaId v_LAI := f_GetLAI(v_PLMN, v_LAC, p_ForcedAttach); // @sic R5s110176 sic@

    var TrackingAreaCode v_TAC := f_EUTRA_CellInfo_GetTAC(p_CellId);

    var template (omit) EPS_NetworkFeatureSupport v_NtwkFeatSupport := p_NtwkFeatSupport;

    var template (omit) EmergNumList v_EmergNumList := omit;

    var template (omit) MobileIdentity v_MSId := f_GetMSId(p_ForcedAttach); // @sic R5-103673 sic@

    var PDN_Index_Type v_PdnIndex := f_EUTRA_InitialRegistrationPdnIndex(p_InitialRegistrationPDN);               // @sic R5-144797 sic@

    var EPS_BearerIdentity v_EpsDefaultBearerId := f_EUTRA_InitialRegistrationEpsBearerId(p_InitialRegistrationPDN);  // @sic R5-144797 sic@

    var template (omit) ESM_Cause v_ESM_Cause := omit;        // @sic R5s120136 sic@

    var template (value) PDN_Address v_PDN_Address;

    var EmergencyNumList v_EmergencyNumList;

    var BEARER_CONTEXT_TYPE v_BearerContextNumber := DEF_1;

    if (pc_IPv6 and pc_IPv4) {              /* @sic R5-120234, R5-120541, R5-120542 sic@ */

      select (p_PDN_TypeToBeUsed) {

        case (forceIPv4only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110010'B); }     // PDN type IPv4 only allowed

        case (forceIPv6only)          { v_ESM_Cause := cs_ESM_Cause_tv('00110011'B); }     // PDN type IPv6 only allowed

        case (pdnTypeAsSupportedByUE) { v_ESM_Cause := omit; }

      }

    }

    // @sic R5s130681 sic@

    select (f_GetTestcaseAttrib_Eutra_Release(testcasename())) { // @sic R5s130508 sic@

      case ("","rel_10") { // rel8 or rel-10 - 36.508 just says rel-9

      }

      case ("rel_9") {

        v_NtwkFeatSupport.imsVoPS := '1'B; // @sic R5s130902 sic@

        v_NtwkFeatSupport.emcBS := '1'B; // @sic R5s130977 sic@

      }

      case else {}

    }

    if (p_NoOfEmergencyNumbers > 0) {

      v_EmergencyNumList := f_Get_EmergencyNumList (p_NoOfEmergencyNumbers, f_EUTRA_MobileInfo_GetLocalEmergencyNumberList()); /* @sic R5s141315 sic@ */

      f_EUTRA_MobileInfo_SetLocalEmergencyNumberList (v_EmergencyNumList); // save the list of locally usable emergency numbers

      v_EmergNumList := f_Build_EmergNumList(p_NoOfEmergencyNumbers, v_EmergencyNumList);

    }

    if (tsc_EUTRA_Registration_PDN1isIMS) {     // @sic R5-144797, R5s141367 sic@

      v_BearerContextNumber := DEF_2;

    }

    v_PDN_Address := f_EUTRA_GetPDNAddress(p_IPv4AllocationViaNasFlag, v_PdnIndex, p_PDN_TypeToBeUsed);

    f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(p_CellId,

                                                                  v_EpsDefaultBearerId,    // @sic R5-144797 sic@

                                                                  tsc_RRC_TI_Def,

                                                                  p_EPS_TI,

                                                                  v_AttachType,

                                                                  v_PDN_Address,

                                                                  v_Guti,

                                                                  cs_TAIListNonConsecutive_lv(v_PLMN, {bit2oct(v_TAC)}),

                                                                  v_LAI,

                                                                  v_MSId,

                                                                  v_ESM_Cause, // @sic R5-104703 sic@

                                                                  p_APN,

                                                                  p_PcoToUE,

                                                                  f_GetAdditionalUpdateResult(p_AdditionalUpdateType, p_ForcedAttach), // @sic R5s110176 sic@

                                                                  p_T3412,

                                                                  p_T3402,

                                                                  p_T3423,

                                                                  p_EquivalentPlmnList,

                                                                  v_EmergNumList,         // @sic R5s130681 sic@

                                                                  v_NtwkFeatSupport,      // @sic R5s130681 sic@

                                                                  -,

                                                                  -,

                                                                  v_BearerContextNumber, // @sic R5-144797 sic@

                                                                  p_EMM_Cause); // @sic R5-145971 sic@

  }

	  function f_EUTRA_508_IMSEmergencyCallLimitedService_Steps7_15(EUTRA_CellId_Type p_CellId,

                                                                NAS_MSG_Indication_Type p_NAS_Ind,

                                                                EUTRA_ATTACH_Type p_ForcedAttach := EPS_Only,   // @sic R5-135069 sic@

                                                                template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v_deact) runs on EUTRA_PTC

  {

    var NAS_MSG_Indication_Type v_NAS_Ind;

    var ProcedureTransactionIdentifier v_EPS_TI;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var EUTRA_SecurityParams_Type v_AuthParams := f_EUTRA_Security_Get();

    var NasCount_Type v_NasCountUL;

    var AccessPointName v_APN;

    var template (omit) ProtocolConfigOptions v_PcoFromUE;

    var template (omit) ProtocolConfigOptions v_PcoToUE;

    v_NAS_Ind := p_NAS_Ind;

    v_PdnConnectivityRequest := v_NAS_Ind.Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;

    // Step 7-8: NAS Security

    v_NasCountUL := f_EUTRA_NAS_ActivateSecurity(p_CellId, v_AuthParams);   // NAS SECURITY COMMAND & COMPLETE

    // Step 9a1 & 2 - ESM information optionally transferred;  @sic R5-110708 sic@

    v_APN := f_ESM_InitialRegistration_InitAPN (p_CellId, f_CheckEsmInfoTransferFlag(v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST), v_EPS_TI, emergencyPDN); // @sic R5s150227 sic@

    // Step 10-11: RRC Security Mode procedures

    v_AuthParams := f_EUTRA_RRC_ActivateSecurity(p_CellId, v_AuthParams, v_NasCountUL); // RRC SECURITY COMMAND & COMPLETE

    f_EUTRA_Security_Set(v_AuthParams); //Saving security parameters

    // Step 12-13 UE Capability check

    f_EUTRA_Capability (p_CellId, tsc_RRC_TI_Def); // This enquires for the eutra capability but does not check the contents of the message received

    // Step 14 - 15 Activate Def Bearer

    v_PcoFromUE := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;  // may be omit

    v_PcoToUE := f_GetDefaultProtocolConfigOptions (v_PcoFromUE); // @sic R5-104703 sic@

    f_EUTRA_IdleUpdated_Step14_15(p_CellId,

                                  p_ForcedAttach,

                                  v_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.addUpdateType,

                                  v_EPS_TI,

                                  f_CheckPCOforIPallocationViaNas (v_PcoFromUE),

                                  v_PcoToUE,

                                  v_APN,

                                  -,

                                  p_T3412, 
                                  -,

                                  -,

                                  -,

                                  -,

                                  -,

                                  -,

                                  emergencyPDN);

    // EXCEPTION: (IP address allocation), steps 1 - 5 of annex C.22 should be performed in parallel

  }


MCC160: Additional changes to be future proven
	  function f_EUTRA_IdleUpdated_Step16_17(EUTRA_CellId_Type p_CellId,

                                         boolean p_ReleaseConnection := true,

                                         EUTRA_DelayForUserPlaneSignalling_Type p_WaitForIMS := waitForIMS,

                                         template (omit) EPS_NetworkFeatureSupport p_NtwkFeatSupport := cs_EPS_NwkFtSup_IMSVoice,

                                         EUTRA_InitialRegistrationPDN_Type p_InitialRegistrationPDN := firstPDN) runs on EUTRA_PTC 

  { /* @sic R5-112600: always delay of 1.2s (i.e. p_WaitForIMS := waitForIMS when calling f_EUTRA_NAS_AttachComplete) sic@ */

    /* @sic R5s130133 change 8: p_WaitForIMS added sic@ */

    var EPS_BearerIdentity v_EpsBearerId := f_EUTRA_InitialRegistrationEpsBearerId(p_InitialRegistrationPDN);  // @sic R5-144797 sic@

    var EUTRA_DelayForUserPlaneSignalling_Type v_WaitForIMS := p_WaitForIMS;

    var boolean v_IMSVoiceSupported := false;

    var boolean v_SecondPdnToBeReleased := false;

    var template (value) B1_Type v_IMSSupport := '0'B;

    if (isvalue(p_NtwkFeatSupport)) {  // @sic R5-130620 sic@

      v_IMSSupport := valueof(p_NtwkFeatSupport.imsVoPS);

      if (match ('1'B, v_IMSSupport)) {

        v_IMSVoiceSupported := true;

      }

    }

    if (not v_IMSVoiceSupported and pc_voice_PS_1_CS_2 and pc_Attach) { // @sic R5-130620 sic@

      v_WaitForIMS := dontWaitForIMS;    // @sic R5-144797: NOTE: acc. to the prose in this case there is no second PDN at all sic@ */

    } else if (tsc_EUTRA_Registration_MultiplePDN) { /* @sic R5-144797: multiple PDN sic@ */

      v_WaitForIMS := noDelay;    /* In case of a second PDN connectivity request this may be sent immediately by the UE

                                     => we cannot wait after the ATTACH_COMPLETE for any trigger from the IMS_PTC but need to do that after the second PDN connectivity is over */

    }

    f_EUTRA_NAS_AttachComplete(p_CellId, v_EpsBearerId, v_WaitForIMS); // step 16  @sic R5s110449 sic@

    if (not v_IMSVoiceSupported and pc_voice_PS_1_CS_2 and pc_Attach) { // @sic R5-130620 sic@

      if (pc_TAU_connected_in_IMS) { // Steps 16a1-3 @sic R5-131455 sic@

        f_EUTRA_ConnectedMode_TAU(p_CellId, -, Combined, tsc_EpsUpdate_Combined_TaLaUpdate);

      }

      else if (pc_TAU_idle_in_IMS) { // Steps 16b1-6 @sic R5-131455 sic@

        f_EUTRA_RRC_ConnectionRelease(p_CellId);

        f_EUTRA_TrackingAreaUpdate_Step1_5 (p_CellId, tsc_RRC_TI_Def, Combined, -, -, tsc_EpsUpdate_Combined_TaLaUpdate);

      }

    } else if (tsc_EUTRA_Registration_MultiplePDN) {   // @sic R5-144797: support of multiple PDN registration sic@

      f_EUTRA_IdleUpdated_ConnectToAdditionalPDN(p_CellId, p_WaitForIMS);

      v_SecondPdnToBeReleased := true;

    }

    if (p_ReleaseConnection) { // @sic R5-105005 Some tests only want to do up to step 16 sic@

      f_EUTRA_RRC_ConnectionRelease(p_CellId);   // step 17  @sic R5s150338: no v_DrbReleaseList anymore sic@

      if (v_SecondPdnToBeReleased) {  // @sic R5-144797: support of multiple PDN registration sic@

        f_EUTRA_IdleUpdated_ReleaseSecondPDN(p_CellId);

      }

    }

  }

	  function f_EUTRA_IdleUpdated_Step5_15(EUTRA_CellId_Type p_CellId,

                                        IDLEUPDATED_STATE_Type p_State,

                                        EUTRA_ATTACH_Type p_ForcedAttach,

                                        NAS_MSG_Indication_Type p_NAS_Ind,

                                        PDN_TypeToBeUsed_Type p_PDN_TypeToBeUsed := pdnTypeAsSupportedByUE,

                                        template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v_deact,

                                        template (omit) GPRS_Timer p_T3402 := omit,

                                        template (omit) GPRS_Timer p_T3423 := omit,

                                        template (omit) PLMN_List p_EquivalentPlmnList := omit,

                                        integer p_NoOfEmergencyNumbers := 0, // @sic R5s130681 sic@

                                        template (omit) EPS_NetworkFeatureSupport p_NtwkFeatSupport := cs_EPS_NwkFtSup_IMSVoice,

                                        template (omit) EMM_Cause p_EMM_Cause := omit,

                                        EUTRA_InitialRegistrationPDN_Type p_InitialRegistrationPDN := firstPDN)

    runs on EUTRA_PTC return APNandPCOs_Type

  { /* @sic R5-110796: split into f_EUTRA_IdleUpdated_Step5_13 and f_EUTRA_IdleUpdated_Step14_15 sic@ */

    /* @sic R5s140560: return value sic@ */

    /* @sic R5-145971: new parameter p_EMM_Cause sic@ */

    var template (omit) AdditionalUpdateType v_AdditionalUpdateType := p_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.addUpdateType;  /* @sic R5s110342 sic@ */

    var NAS_UL_Message_Type v_PdnConnectivityRequest := p_NAS_Ind.Pdu.PiggybackedPduList[0].Msg;

    var ProcedureTransactionIdentifier v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;

    var template (value) ProtocolConfigOptions v_PcoToUE;

    var template (omit) ProtocolConfigOptions v_PcoFromUE := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;  // may be omit

    var boolean v_IPv4AllocationViaNasFlag := f_CheckPCOforIPallocationViaNas(v_PcoFromUE); // See TS 36.508 v.8.2.2 Table 4.7.3-6

    var APNandPCOs_Type v_APNandPCOs := f_EUTRA_IdleUpdated_Step5_13(p_CellId, p_State, p_NAS_Ind);  // @sic R5s130382 sic@

    var PDN_Index_Type v_PdnIndex := f_EUTRA_InitialRegistrationPdnIndex(p_InitialRegistrationPDN);            // @sic R5-144797 sic@

    if (ispresent(v_APNandPCOs.protocolConfigurationOptions)) { // @sic R5-133116 sic@

      v_PcoFromUE := v_APNandPCOs.protocolConfigurationOptions;

      v_IPv4AllocationViaNasFlag := v_IPv4AllocationViaNasFlag and f_CheckPCOforIPallocationViaNas(v_PcoFromUE);

    }

    v_PcoToUE := f_GetDefaultProtocolConfigOptions(v_PcoFromUE, v_PdnIndex); // @sic R5-104703 sic@

    f_EUTRA_IdleUpdated_Step14_15(p_CellId,

                                  p_ForcedAttach,

                                  v_AdditionalUpdateType,

                                  v_EPS_TI,

                                  v_IPv4AllocationViaNasFlag,

                                  v_PcoToUE,

                                  v_APNandPCOs.accessPointName,

                                  p_PDN_TypeToBeUsed,

                                  p_T3412,

                                  p_T3402,

                                  p_T3423,

                                  p_EquivalentPlmnList,

                                  p_NoOfEmergencyNumbers,

                                  p_NtwkFeatSupport,

                                  p_EMM_Cause,

                                  p_InitialRegistrationPDN);
    return v_APNandPCOs;  /* @sic R5s140560 sic@ */

  }


